Pulse Programming in TopSpin 3.1.5

Presented by Mike Brown

04152012

What we will cover here

- Basic of a Bruker Pulse Program
- Some Bruker Conventions
- Pulse Powers and “edprosol”
- Changes to Decoupling schemes

Basic Features of a Bruker Pulse Program

The first few line will contain the title, date, type, literature references and some other information about the pulse program

;deptq

;avance-version (05/06/10)

;dept polarization transfer

;with decoupling during acquisition

;with fixed 135 degree read pulse

;R. Burger & P. Bigler, JMR. 135, 529-534 (1998)

Basic Features of a Bruker Pulse Program

Then you will find include files and definitions

```
#include <Avance.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d2=1s/(cnst2*2)"
"d12=20u"
"p0=(p3/90)*135"

"DELTA=p1*4/3.1416"
```

Basic Features of a Bruker Pulse Program

Then the main body of the pulse program

```
1 ze
2 30m do:f2
  d1
  d12 pl2:f2
  (p1 ph2 d2):f1
  (p2 ph3 d2):f1 (p3 ph1):f2
  (p1 ph1 d2):f1 (p4 ph3):f2
  (p2 ph5 d2):f1 (p0 ph4):f2
  DELTA pl12:f2
  go=2 ph31 cpd2:f2
  30m do:f2 mc #0 to 2 F0(zd)
exit
```

Basic Features of a Bruker Pulse Program

Then we have the phasing programs for the various pulses and receiver

```
ph1=0
ph2=3
ph3=0 2 1 3
ph4=1
ph5=0 2
ph31=1 1 3 3
```



Basic Features of a Bruker Pulse Program

Then we have the comments section

;pl1 : f1 channel - power level for pulse (default)

;pl2 : f2 channel - power level for pulse (default)

;pl12 : f2 channel - power level for CPD/BB decoupling

;p0 : f2 channel - 135 degree - X, XH2 negative and XH, XH3 positive,

;p1 : f1 channel - 90 degree high power pulse

;p2 : f1 channel - 180 degree high power pulse

;p3 : f2 channel - 90 degree high power pulse

;p4 : f2 channel - 180 degree high power pulse

;d2 : 1/(2 * J(XH))

;d12 : delay for power switching [20 usec]

Basic Features of a Bruker Pulse Program

```
;cnst2 : = J(XH)
;NS    : 4 * n, total number of scans: NS * TD0
;DS    : 8
;cpd2  : decoupling according to sequence defined by cpdprg2
;pcpd2 : f2 channel - 90 degree pulse for decoupling sequence
;DELTA : delay to compensate for chemical shift evolution during
;         RF pulse in order to give same phase correction as
;         for reference 1D spectrum
```

Basic Features of a Bruker Pulse Program

You can use “showpp” or “spdisp” to view the timing of the pulse program.



Some conventions that will always hold true for Bruker Pulse Programs

- p1 : Observe Nucleus 90 degree pulse width (at plw1)
- p2 : Observe Nucleus 180 degree pulse width (at plw1)
- p3 : Decouple Nucleus 90 degree pulse width (at plw2)
- p4 : Decouple 180 degree pulse width (at plw2)
- d1 relaxation delay; 1-5 * T1
- d2 : 1/(2JXH) calculated from cnst2
- d3 : 1/(3JXH) calculated from cnst2
- d4 : 1/(4JXH) calculated from cnst2
- d5 : DE/2
- d6 : delay for evolution of long range couplings
- d7 : delay for inversion recovery
- d8 : NOESY mixing time
- d9 : TOCSY mixing time

Pulse Powers

- In TopSpin 3 Bruker introduced the concept of watts and –dBw for calculating powers.
- The easiest way to work with the powers in TopSpin 3
  - Look in edhead find out the maximum power for each coil in your probe
  - Make sure POWCHECK is on
  - Never try to put too much power into your probe.
  - The shortest Proton 90 should be about 6 usec. (conventional)
  - The shortest Carbon 90 should be about 9 usec. (conventional)
  - Increase the power, measure the 90, and get it as short as you can without exceeding the power limitations of the probe
  - In edprosol make sure the powers are in watts

edprosol

Select
Power/Watt



Pulse Powers

- Remember that the default pulse power for channel 1 is plw1
- And the default for channel 2 is plw2
- You can change this behavior by adding a line to the pulse program such as

  d11 pl14:f2

  This would set the power level of pl14 to channel f2.

Decoupling

- Decoupling must start during a delay of at least 4 usec or more
- The statement cpd:f2 will turn on cpd decoupling on channel f2

  d11 pl12:f2

  d11 cpd2:f2
- d11 pl12:f2 will set the power level pl12 to the f2 channel
- d11 cpd2:f2 will turn on cpd decoupling on the f2 channel
- Later in the pulse program the statement do:f2 will be encountered, (after a delay) this will turn the decouple off at that time
- The time for all of these commands is d11, usually d11 is about 30 milliseconds.

Some rules for Pulse programing in TopSpin 3.1

- The statement :

  (p1 ph1):f1

  100u (this time can be as short as 0 usec)

  (p2 ph2):f2

- Executes a pulse on channel f1, followed by a delay, followed by a pulse on f2



Some rules for Pulse programing in TopSpin 3.1

- The statement:

  (p1 ph1 100u):f1

  (p2 ph2):f2

- Executes the pulse on channel f1, followed by a delay, followed by a pulse on channel f2



Some rules for Pulse programing in TopSpin 3.1

- The statement:

  (p1 ph1):f1 (100u)

  (p2 ph2):f2

- Executes a pulse on channel f1 At the same time as the 100 usec delay starts. The pulse on f2 does not start until either p1 or the 100 usec delay has passed, whichever is longer.



Some rules for Pulse programing in TopSpin 3.1

- The following is an example from a typical section of a DEPT pulse program:

  (p4 ph2):f2 (p1 ph4 d2):f1

  (p0 ph3):f2 (p2 ph5 d2):f1

- The pulses p4 and p1 begin at the same time
- The pulses p0 and p2 start simultaneously, but not before the pulses of the previous line finishes.



Some rules for Pulse programing in TopSpin 3.1

- Individual pulse alignment
- Use the statement refalign to indicate the reference

```
(
  refalign (d0 p1 ph1 d0):f1
  center (p2 ph2):f2
  ralign (p3 ph4):f3
)
```

- The f3 channel is right-aligned to f1
- f2 channel is centered relative to f1



CPD decoupling

- The statement **cpd** will continue from where it was stopped the last scan at the command do
- The statement **cpds** will start the decoupling sequence from line 1 of the decoupling scheme
- The statement **cpdngs** or **cpdng** is the same as above but the decoupler gate for the channel will not be opened, gating is controlled by the main pulse program.
- New CPD sequences:
  - **waltz64.p61**, uses p61, for triple resonance experiments
  - **garp4.p61,** uses p61, for triple resonance experiments
  - There are some new adiabatic and bilevel sequences
  - There are also some new random phase sequences for polymer work: waltz17, waltz 65

CPD decoupling

- Features for the AV III only:
  - Phase programs for the CPD programs can be used and defined in the same way as the main pulse program.

    An example:

    1 p15:sp15 ph21^

    jump to 1

    ph21=(360) 0 15 165 180
  - The phase pointer of ph21 can also be manipulated in the main program as well with the command **ipp21**

New and improved Pulse programs and Parameter sets

- Quite a few new pulse programs were added for triple resonance 13C detected. Such as: **c_hcacon_ia , c_hcacon_ia.2, c_hcacon_ia3d, c_hcacon_ia3d.2, c_hcacon_iarc_caco** there are quite few more.

- Some pulse programs have been corrected, one in particular is the **adeq11etgpjcrdsp**, pl2:f2 added (adequate, echo anti-echo)

- There are some new pulse programs for triple resonance, NOESY, Relaxation HSQC, and Relaxation experiments that incorporate temperature compensation, HMBC, HSQC and 2 new inadequates.

- New APSY parameter sets are included along with the appropiate pulse programs.

Base-opt

- A feature only available on the AV III (IPSO) is **DIGMOD baseopt**
- To use this feature the last pulse before the acquisition in the pulse program must be a 90 degree pulse

And

- You must have the statement

  "acqt0=-p1*2/3.1416"

  After the include statement in the pulse program
- And you must make digmod = baseopt. (in eda)
- This will remove those "smiles" from the spectra and decrease substantially the need for first order phase correction.
- Some canned pulse programs will have the statement already incorporated some will not.

Base-opt



Without base-opt



With base-opt

New Gradients in TopSpin 3

- Square gradients have been changed to smooth square gradients
  - Such as **SMSQ10.10, SMSQ10.32, SMSQ10.100**
  - The 10 means there is 10% smoothing

- The SINE gradients now have a non-zero first point and a zero last point

- All of the gradients have been recreated with the latest Shapetool version

New Shaped pulses and changes in TopSpin 3.1

- A new shaped pulse was added Iburp2.1000
- The integral factor was changed in Crp32,1.5,20.2, Crp42,1.5,20.2, Crp8,1.5,20.2 so the power applied can be dropped by 2.22 dB
- Gaussramp-down.1, Gaussramp+down.1, Gaussramp-up.1, Gaussramp+up.1 were added for EASY ROESY, You can manipulate the spinlock axis, with a + or – offset, either up or down.
- All of the Q3_surbop.1, Q5_sebop.1, Q5tr_sebop.1, Burbop-180.1 shaped pulses have been optimized for less phase error.

New Shaped pulses and changes in TopSpin 3.1

- A special adiabatic pulse has been added and optimized:

  Crp60,20,20.0.1, 60kHz, 20 millisecond, 20% smoothed, 1000 points

  Chirp pulse for use in the ZQ filter

- You can find out about changes in each version by reading the update.info file in each directory (wave, cpd, pp, and gp)

Are there any questions?

28



www.bruker-biospin.com